TP3 : Vecteurs, matrices, systèmes linéaires et boucles

Création de matrices

Création d'un vecteur ligne

Essayer les commandes suivantes, et répondre aux questions.

```
--> x = [-2 -1  0  1  2]
--> y = [ 0  1 -1 3  1]
--> a = [x y]            // Quel est le vecteur [x y]?
--> v = 2:0.1:3          // Quelles sont les première et dernière valeurs ?
--> w = 2:0.3:3          // Quelle est la raison ? Pourquoi ça ne termine pas par 3 ?
--> z = 2:6              // Quelle est la raison par défaut ?
--> c = linspace(0,4,5)  // Quelle est la différence
--> d = linspace(0,4,7)  // entre a:b:c et linspace ?
```

Création d'un vecteur colonne.

Les lignes sont séparées par des points-virgules.

```
-->v = [1;2;3;4]          // Crée un vecteur unicolonne à quatre composantes.
```

Création d'une matrice carrée

```
-->A = [1 2;3 4]
```

On peut produire une nouvelle matrice en effectuant des opérations sur une autre matrice :

```
-->log(A)
```

Exercice 1 Créer la matrice $B = \begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{pmatrix}$.

Transposition

```
-->B
-->B'               // Que fait le ' ?
```

Exercice 2 Créer le vecteur colonne $V = \begin{pmatrix} 1 \\ 2 \\ 3 \\ 4 \end{pmatrix}$ en utilisant le '.

La commande $eye(2,2)$ crée la matrice unité $\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$.

Que fait alors la commande $eye(3,3)$?

Exercice 3 Trouver au moins une commande pour définir les vecteurs $u = (1\ 2\ 3\ 4)$, $v = \begin{pmatrix} 1 \\ 2 \\ 3 \\ 4 \end{pmatrix}$ et la matrice $C = \begin{pmatrix} 1 & 1 \\ 2 & 2 \\ 3 & 3 \end{pmatrix}$.

Accès aux valeurs
Si $A = (a_{i,j})$, le coefficient situé à l'intersection de la i-ème ligne et de la j-ème colonne s'obtient par l'instruction $A(i,j)$:

```
-->A
-->A(2,1)              // On peut le lire
-->A(2,1)=9            // On peut le modifier
-->A
```

Boucles

Une *boucle* est une commande qui ordonne à Scilab d'exécuter plusieurs fois la même instruction. Par exemple :

```
--> for x=1:5
-->   x^2
--> end;
```

affiche les carrés des entiers de 1 à 5.

On peut imbriquer des boucles :

```
--> for i=1:3
-->   for j=1:3
-->      i*j
-->   end;
--> end;
```

On peut utiliser des boucles pour calculer les coefficients de la matrice :

```
--> for i=1:3
-->   for j=1:3
-->      A(i,j) = i+j;
-->   end;
--> end;
--> A
```

Exercice 4 Écrire une commande avec des boucles qui crée une matrice carrée A de taile 5×5 telle que $a_{i,j} = i^2 j$.

Opérations matricielles

Commencer par définir les deux matrices $A = \begin{pmatrix} 1 & 3 \\ 2 & -1 \end{pmatrix}$ et $B = \begin{pmatrix} -1 & 1 \\ -2 & 3 \end{pmatrix}$.

Addition:

```
--> A+B
```

Une autre façon de faire est d'additionner les composantes une à une en utilisant des boucles :

```
--> for i = 1:2
-->   for j = 1:2
-->     C(i,j) = A(i,j) + B(i,j);
-->   end;
--> end;
--> C
```

Multiplication:

```
--> A*B     // et que fait A.*B?
```

On peut aussi revenir à la définition : si $C = AB$, alors $C_{ij} = \sum_{k=1}^{2} A_{ik}B_{kj}$.

```
--> for i = 1:2
-->   for j = 1:2
-->     C(i,j) = A(i,1) * B(1,j) + A(i,2)*B(2,j);
-->   end;
--> end;
```

Exercice 5 Essayer de calculer le produit d'une matrice 3×2 avec une autre matrice 3×2. Qu'arrive t-il ? Expliquer. Même question avec l'addition d'une matrice 3×2 et d'une matrice 2×2.

Exercice 6 Soit A la matrice $A = \begin{pmatrix} -1 & 0 & 1 \\ 0 & 0 & 3 \\ 2 & 0 & 0 \end{pmatrix}$. Trouver les commandes pour calculer $B = A^4$ et C la matrice dont les composantes sont les composantes de A à la puissance 4.

Résolution des systèmes d'équations linéaires

Résolution de l'équation $Ax = b$:
Un système d'équations linéaires se traduit matriciellement par $Ax = b$ où A est une matrice, b un vecteur colonne et x le vecteur colonnes d'inconnues. Ainsi, résoudre $Ax = b$, c'est trouver tous les vecteurs x qui satisfont cette équation. Pour la résolution d'une telle équation, on se limitera au cas où A **est une matrice carrée**. Dans ce cas particulier, la matrice A est en général inversible, mais pas toujours ! Si elle est inversible, alors le système a une unique solution, donnée par $x = A^{-1}b$.
Pour le calcul de l'inverse d'une matrice A, il suffit de taper

```
--> A^(-1)
```

Essayer les commandes suivantes :

```
--> A=[1 2;1 1];
--> A^(-1)
--> A^(-1)*A                // Vérification !
--> A.^(-1)                 // Que fait A.^(-1) ?
--> A.^(-1) .* A            // Vérification !
--> b=[3;5]
--> x=A^(-1)*b              // Résolution de l'équation Ax=b
--> A=[1 2 3;4 5 6]
--> A^(-1)                  // Que se passe-t-il?

--> A=[1 1;2 2]
--> A^(-1);                 // Cette matrice A n'est pas inversible
--> linsolve(A,b)           // b est défini plus haut : pas de solution de Ax+b=0
--> b=[1;2]
--> x0= linsolve(A,b)       // x0 est une solution de Ax+b=0
--> y= kernel(A)            // y est une solution de Ax=0
--> [x0,y]=linsolve(A,b)    // les solutions de Ax+b=0 sont de la forme x0+ty avec t réel.
```

Exercice 7 Résoudre matriciellement les systèmes linéaires suivants avec Scilab :

1. $\left\{\begin{array}{ccccccc} x & + & y & + & z & = & 1 \\ & & 2y & + & 3z & = & 2 \\ x & - & y & & & = & 0 \end{array}\right.$,

2. $\left\{\begin{array}{ccccccc} x & - & y & + & z & = & 0 \\ 2x & & & + & z & = & 1 \\ 3x & + & y & + & z & = & 2 \end{array}\right.$ puis $\left\{\begin{array}{ccccccc} x & - & y & + & z & = & 0 \\ 2x & & & + & z & = & 1 \\ 3x & + & y & + & z & = & 0 \end{array}\right.$.

Attention à bien résoudre $Ax - b = 0$ et à ne pas résoudre $Ax + b = 0$.

Calcul de l'inverse d'une matrice par pivot de Gauss

Recopier le code suivant :

```
--> function A = AjLigne(A,x,i,j)
--> [m,n] = size(A)   // Calcul de la taille de A
--> for k=1:n
--> A(j,k) = A(j,k) + x * A(i,k)
--> end
--> endfunction
```

Ce code crée une fonction appelée AjLigne qui prend une matrice A, un nombre réel x et deux entiers i et j, qui ajoute x fois la ligne i à la ligne j de A, et qui renvoie la matrice A modifiée (mais la matrice A n'est pas elle-même modifiée par cette fonction).

Exercice 8 En s'inspirant de ce code, écrire une fonction Multligne(A,i,x) qui multiplie la ligne i d'une matrice A par x.

Le code suivant crée une matrice carrée A et lui accole une matrice identité :

```
--> A = [1 2; 2 3]
--> B = [A eye(2,2)]
```

Exercice 9 En utilisant les fonctions AjLigne et MultLigne, calculer l'inverse de A par pivot de Gauss.

Exercice 10 Faire de même avec la matrice $A = \begin{pmatrix} 0 & 1 & 1 \\ 1 & 0 & 1 \\ 1 & 1 & 0 \end{pmatrix}$.